#===============================================
#global setting
#===============================================


DEFINES += RIVIADEBUG

#配置生成的可执行程序目录
CONFIG(debug, debug|release){
    contains(QT_ARCH, x86_64){
        DESTDIR = $$PWD/../build/bin/X64/Debug
    } else {
        DESTDIR = $$PWD/../build/bin/X86/Debug
    }
}

CONFIG(release, debug|release){
    contains(QT_ARCH, x86_64){
        DESTDIR = $$PWD/../build/bin/X64/Release
    } else {
        DESTDIR = $$PWD/../build/bin/X86/Release
    }
}

#配置生成的应用程序名，即软件名
unix{
    CONFIG(debug, debug|release){
        contains(QT_ARCH, x86_64){
            TARGET=$$join(TARGET,,,_L64d)
        } else {
            TARGET=$$join(TARGET,,,_L32d)
        }
    }

    CONFIG(release, debug|release){
        contains(QT_ARCH, x86_64){
            TARGET=$$join(TARGET,,,_L64)
        } else {
            TARGET=$$join(TARGET,,,_L32)
        }
    }
}
win32{
    CONFIG(debug, debug|release){
        contains(QT_ARCH, x86_64){
            TARGET=$$join(TARGET,,,_W64d)
        } else {
            TARGET=$$join(TARGET,,,_W32d)
        }
    }

    CONFIG(release, debug|release){
        contains(QT_ARCH, x86_64){
            TARGET=$$join(TARGET,,,_W64)
        } else {
            TARGET=$$join(TARGET,,,_W32)
        }
    }
}
